David Michel
david.michel@ens-rennes.fr
perso.eleves.ens-rennes.fr/people/David.Michel

ECE 1
Lycée Saint-Vincent
2016-2017

TP4 : Exercices – Corrigé

Ce corrigé présente sommairement les solutions, c'est-à-dire les fonctions et programmes corrects qui répondent aux questions. Pour plus de détails, je vous renvoie aux notes prises pendant les TP. En cas de question, ne pas hésiter à m'envoyer un mail.

Q.1 a. On peut écrire

```
n=input("Entrer un entier naturel : ")
u=1;
for k=1:n
      u=1/(1+u);
end
disp(u)
```

b. On obtient $u_{10} = 0,6180556$.

c. On obtient $u_{20} = 0,6180340$.

d. On a $u_{10} \approx 0,6180556$ et $u_{20} \approx 0,6180340$ et $u_{30} \approx 0,6180340$ et $u_{40} \approx 0,6180340$. On voit que les termes semblent être tous égaux à partir d'un certain rang. En fait, ils sont distincts mais trop proche pour la précision avec laquelle Scilab calcule par défaut. Quoiqu'il en soit, on peut conjecturer que la suite $(u_k)_k$ converge (et on le montre très facilement) vers un réel ℓ dont une valeur approchée à 10^{-2} près est $0,62$.

Q.2 On doit calculer les termes successifs de la suite, tant que $u_n < 10^5$, on utilise donc une boucle **while**. De plus, on veut savoir à quel rang de la suite on s'est arrêté, c'est-à-dire combien de termes on a calculé : on garde donc en mémoire un compteur :

```
function n=premier(u0)
    u=u0
    n=0
    while u<10^5
        u=u+log(u)
        n=n+1
    end
endfunction
```

Q.3 On va à nouveau devoir utiliser une boucle while. De plus, la condition d'arrêt fait intervenir deux termes successifs de la suite. Il faut donc toujours garder en mémoire deux termes consécutifs.

```
epsilon=input("Entrer un réel >0 ")
u=0;
v=exp(-u);       //v jouera le rôle de u_n et u celui de u_{n-1}
while abs(u-v)>epsilon
    u=v;
    v=exp(-v);
end
disp("u(n-1)="+string(u)+" et u(n)="+string(v))
```

Q.4 Comme précédemment, on utilise une boucle **while** et la condition d'arrêt fait intervenir deux termes consécutifs de la suite. De toute façon, il s'agit d'une suite récurrente d'ordre 2 donc, pour en calculer les termes, il est nécessaire de garder en mémoire deux termes consécutifs et utiliser une variable auxiliaire pour mettre à jour correctement **u** et **v** :

```
epsilon=input("Entrer un réel >0 ")
u=1;
v=1/2;
while u-v>epsilon
    aux=u;
    u=v;
    v=v*(1+v-aux);
end
disp(v)
```